Condicionales y Funciones

Prof.: Dr. Alejandro S. M. Santa Cruz
J.T.P.: Dr. Juan Ignacio Manassaldi
Aux. 1ra: Ing. Amalia Rueda

Condicionales (if)



if: palabra clave para ejecución condicional

Ejercicio

```
if expr1 then
    statements
elseif expri then
    statements
    ....
else
    statements
end
```



Condicionales

```
a=4;
if a>2 then
        disp('a es mayor que 2');
elseif a==2 then
        disp('a es igual a 2');
else
        disp('a es menor que 2');
end
```



Condicionales

```
a=[-1 2 3 -6 -8 0 1]
j=1;
k=1;
for i=1:length(a)
    if a(i)>=0 then
        p(j)=a(i);
        j=j+1;
    else
        n(k)=a(i);
        k=k+1;
    end
end
```



Condicionales


```
while condición
    bloque
end
```

while: palabra clave de ingreso a un bucle condicional

Condicionales


```
a=[1 2 3 -6 -8 0 1];
p=1;
while a(p)~=0
    p=p+1;
end
p
```

Condicionales


```
a=[1 2 3 -6 -8 0 1]
for i=1:length(a)
    if a(i)==0 then
        p=i
        break
    end
end
```

break: palabra clave para interrumpir bucles

Condicionales (select)

```
select variable
case value1 then
    instructions 1
case value2 then
    instructions 2
    ...
case valuen then
    instructions n
[else     instructions]
end
```

Ejecuta un conjunto de instrucciones según el valor que tome la variable seleccionada

Funciones personalizadas

```
function [salidas]=nombre(entradas)
//Se desarrolla la función de manera que se puedan
//calcular todas las salidas a partir de los valores de
//entrada.
endfunction
```

```
function [media]=media_aritmetica(x)
//Calcula la media aritmetica del vector ingresado
media=sum(x)/length(x);
endfunction
```

Funciones: Ejemplo I

- Densidad del aire como gas ideal

```
function [densidad]=dens_ideal(P, T)
// P en atmosfera
// T en ºC
// PM en g/mol
// R en atm.L/mol.ºK
// densidad en g/L
Tk=T+273.15;
PM=29;
R=0.082;
densidad=(P*PM)/(R*Tk);
endfunction
```

```
--> rho=dens_ideal(1,25)
 rho   =

   1.1861765
```

Funciones: Ejemplo II

- Temperatura de saturación del agua

```
function T4_p=Tsat_p(p)
//Release on the IAPWS Industrial formulation 1997
//for the Thermodynamic Properties of Water and Steam, September 1997
//Section 8.2 The Saturation-Temperature Equation
//Eq 31, Page 34
//p MPa
//Tsat K
beta = p ^ 0.25;
E = beta ^ 2 - 17.073846940092 * beta + 14.91510861353;
f = 1167.0521452767 * beta ^ 2 + 12020.82470247 * beta - 4823.2657361591;
G = -724213.16703206 * beta ^ 2 - 3232555.0322333 * beta + 405113.40542057;
D = 2 * G / (-f - (f ^ 2 - 4 * E * G) ^ 0.5);
T4_p = (650.17534844798 + D - ((650.17534844798 + D) ^ 2 ..
- 4 * (-0.23855557567849 + 650.17534844798 * D)) ^ 0.5) / 2;
endfunction
```

```
--> NBP=Tsat_p(0.101325)
 NBP  =

   373.12430
```

Funciones: Ejemplo III

- Función y sus derivadas

```
function y=fun(x, deriv)
select deriv
        case 0 then
                y=2*x.^3+5*x.^2;
        case 1 then
                y=6*x.^2+10*x;
        case 2 then
                y=12*x+10;
        case 3 then
                y=12;
        else
                y=0;
        end
endfunction
```

```
--> fun(2,0)
 ans  =

   36.
```

```
--> fun(2,1)
 ans  =

   44.
```

```
--> fun(2,4)
 ans  =

   0.
```

Ejercicio



Encontrar la expresión general de cada elemento

Ejemplo

```
A=[1 2 5
   7 6 7
   3 8 4
   1 3 6];
B=[3 1 7 2
   4 8 0 4
   6 2 1 6];

[m p]=size(A)
[q n]=size(B)

for i=1:m
    for k=1:n
        suma=0;
        for j=1:p
            suma=suma + A(i,j)*B(j,k);
        end
        P(i,k)=suma;
    end
end
```

Ejemplo

```
function P=Pmatriz(A,B)
[m p]=size(A);
[q n]=size(B);
if p==q
    for i=1:m
        for k=1:n
            suma=0;
            for j=1:p
                suma=suma + A(i,j)*B(j,k);
            end
            P(i,k)=suma;
        end
    end
else
    disp('las dimensiones no son correctas');
end
endfunction
```